3. Coordinate omogenee e trasformazioni dello spazio

Passiamo ora a considerare le trasformazioni dello spazio tridimensionale. Lo spazio sarà identificato, mediante l'introduzione di un sistema di riferimento cartesiano, con lo spazio vettoriale R^3 delle terne ordinate (x, y, z) di numeri reali.

3.1 Coordinate omogenee

Le coordinate omogenee dei punti dello spazio possono essere introdotte in maniera simile a quanto visto nel caso del piano. Un punto di coordinate cartesiane (x, y, z) ha <u>*coordinate omogenee*</u> una qualsiasi quaterna (X, Y, Z, W) di R^4 tale che $\mathrm{W} \neq 0$ e $\frac{X}{\mathrm{W}} = x, \frac{Y}{\mathrm{W}} = y, \frac{Z}{\mathrm{W}} = z.$

Dunque $(x, y, z, 1)$ e ogni suo multiplo $(r\,x, r\,y, r\,z, r)$ sono coordinate omogenee dello stesso punto (x, y, z) dello spazio. Un 'punto' con coordinate omogenee (non tutte nulle) $(X, Y, Z, 0)$ non corrisponde a un punto dello spazio tridimensionale, ma rappresenta un <u>*punto all'infinito*</u> nella direzione del vettore tridimensionale (X, Y, Z). L'insieme costituito da tutte le quaterne non nulle (X, Y, Z, W), in cui due quaterne vengono identificate se sono una multiplo dell'altra, forma lo <u>*spazio proiettivo tridimensionale*</u> P^3.

Esempio: le coordinate omogenee (5, –2, 1, 2) e (–10, 4, –2, –4) rappresentano lo stesso punto dello spazio di coordinate cartesiane $\left[\frac{5}{2}, -1, \frac{1}{2}\right]$.

3.2 Trasformazioni dello spazio

Una <u>*trasformazione affine*</u> dello spazio è un'applicazione ottenuta componendo un' <u>*applicazione lineare*</u> di R^3 con una <u>*traslazione*</u>:

$$T = T_{v_0} \circ L$$

con L applicazione lineare determinata da una matrice A di ordine 3 e $v_0 = (h, k, l)$ un vettore che definisce la traslazione T_{v_0}. La traslazione trasforma il punto di coordinate cartesiane (x, y, z) nel punto $(x + h, y + k, z + l)$. T è definita da $\mathrm{T}(x, y, z) = (x_1, y_1, z_1)$, con

$$[x_1 \quad y_1 \quad z_1] = [x \quad y \quad z]\,.A + [h \quad k \quad l]$$

In coordinate omogenee, l'equazione precedente si riscrive mediante un unico prodotto matriciale con una matrice 4x4:

$$[X_1 \quad Y_1 \quad Z_1 \quad \mathrm{W}_1] = [X \quad Y \quad Z \quad \mathrm{W}] \begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} & 0 \\ a_{2,1} & a_{2,2} & a_{2,3} & 0 \\ a_{3,1} & a_{3,2} & a_{3,3} & 0 \\ h & k & l & 1 \end{bmatrix}$$

Infatti, tale prodotto equivale a

$$[X_1 \quad Y_1 \quad Z_1] = [X \quad Y \quad Z]\,A + [h\,\mathrm{W} \quad k\,\mathrm{W} \quad l\,\mathrm{W}] \quad \text{e} \quad \mathrm{W}_1 = \mathrm{W}$$

cioè, dividendo a sinistra per W_1 e a destra per W,

$$\left[\frac{X_1}{\mathrm{W}_1} \quad \frac{Y_1}{\mathrm{W}_1} \quad \frac{Z_1}{\mathrm{W}_1}\right] = \left[\frac{X}{\mathrm{W}} \quad \frac{Y}{\mathrm{W}} \quad \frac{Z}{\mathrm{W}}\right] A + [h \quad k \quad l].$$

Traslazioni

La matrice di trasformazione è

$$\mathrm{T}(h, k, l) = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ h & k & l & 1 \end{bmatrix}$$

Esempio:

```
> with(linalg):
> T3d:=(x0,y0,z0)->matrix(4,4,[[1,0,0,0],[0,1,0,0],[0,0,1,0],[x0,y0,z0,1]]):
> trasforma:=(f,m)->delcols(evalm(augment(f,vector(rowdim(f),1))&*m),rowdim(m)..rowdim(m)):
> trasforma([[x,y,z]],T3d(h,k,l));
>
```

Cambiamenti di scala

La matrice corrispondente ai *fattori di scala* s_x, s_y, s_z è

$$\mathrm{S}(s_x, s_y, s_z) = \begin{bmatrix} s_x & 0 & 0 & 0 \\ 0 & s_y & 0 & 0 \\ 0 & 0 & s_z & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Osservazione: in particolare, scegliendo fattori di scala 1,1 e -1, nell'ordine opportuno, si ottengono le matrici che rappresentano le *riflessioni* rispetto ai piani coordinati *x y*, *x z* e *y z*:

$$R_{yz} = \mathrm{S}(-1, 1, 1),\ R_{xz} = \mathrm{S}(1, -1, 1),\ R_{xy} = \mathrm{S}(1, 1, -1)$$

Esempio

```
> S3d:=(sx,sy,sz)->matrix(4,4,[[sx,0,0,0],[0,sy,0,0],[0,0,sz,0],[0,0,0,1]]):
> trasforma([[x,y,z]],S3d(2,3,4));
>
```

Rotazioni primarie

Nello spazio si possono considerare le rotazioni attorno a una retta, chiamata *asse di rotazione*. Le *rotazioni primarie* sono quelle attorno a un'asse coordinato, con angolo di rotazione positivo determinato dalla regola della "vite destrorsa": una vite che punta nella direzione positiva di un asse coordinato, avanza verso tale direzione quando l'angolo di rotazione è positivo. Ad esempio, una rotazione di $\frac{\pi}{2}$ attorno all'asse *y* porta i punti dell'asse *z* sull'asse *x*.

Le matrici che rappresentano le rotazioni primarie si ottengono facilmente dalla matrice che rappresenta una rotazione nel piano:

$$Rot_x(\theta) = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & \cos(\theta) & \sin(\theta) & 0 \\ 0 & -\sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix},\quad Rot_y(\theta) = \begin{bmatrix} \cos(\theta) & 0 & -\sin(\theta) & 0 \\ 0 & 1 & 0 & 0 \\ \sin(\theta) & 0 & \cos(\theta) & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix},$$

$$Rot_z(\theta) = \begin{bmatrix} \cos(\theta) & \sin(\theta) & 0 & 0 \\ -\sin(\theta) & \cos(\theta) & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Esempio

```
> Rotx:=t->matrix(4,4,[[1,0,0,0],[0,cos(t),sin(t),0],[0
  ,-sin(t),cos(t),0],[0,0,0,1]]):
> Roty:=t->matrix(4,4,[[cos(t),0,-sin(t),0],[0,1,0,0],[
  sin(t),0,cos(t),0],[0,0,0,1]]):
> Rotz:=t->matrix(4,4,[[cos(t),sin(t),0,0],[-sin(t),cos
  (t),0,0],[0,0,1,0],[0,0,0,1]]):
> Rotx(Pi/3);Roty(Pi);Rotz(Pi/4);
>
```

Rotazione con asse arbitrario

Se l'asse di rotazione non è un asse coordinato, la rotazione può essere ottenuta trasformando l'asse di rotazione in uno degli assi coordinati, per esempio l'asse *z*, applicando poi la rotazione primaria attorno all'asse *z*, e infine componendo con la trasformazione che riporta l'asse *z* nell'asse di rotazione.

Si noti che per definire il verso della rotazione attorno all'asse, è necessario fissare un orientamento lungo l'asse di rotazione (esattamente come per gli assi coordinati). Questo può essere fatto scegliendo un vettore direzione per l'asse di rotazione.

Sia dunque *r* l'asse di rotazione, passante per un punto $P_0 = (x_0, y_0, z_0)$ e con direzione un vettore $v = (v_1, v_2, v_3)$ che possiamo supporre unitario: $|v|^2 = v_1^2 + v_2^2 + v_3^2 = 1$. Il vettore *v* definisce un punto sulla sfera di raggio1 centrata nell'origine. Sulla sfera è possibile introdurre due *coordinate sferiche*, la *longitudine* θ e la *colatitudine* ϕ, che sono rispettivamente l'angolo tra l'asse *x* e la proiezione ortogonale di *v* sul piano *xy* e l'angolo tra l'asse *z* e il vettore *v*.



La proiezione di v sul piano xy ha lunghezza $\sin(\phi)$ e dunque $v_1 = \sin(\phi)\cos(\theta)$ e $v_2 = \sin(\phi)\sin(\theta)$. Inoltre $v_3 = \cos(\phi)$.
La concatenazione

$$\mathrm{T}(-x_0, -y_0, -z_0)\, Rot_z(-\theta)\, Rot_y(-\phi)$$

trasforma l'asse r di rotazione nell'asse z, poiché porta il punto P_0 nell'origine e il vettore direzione v nel vettore (0, 0, 1). Per ottenere la rotazione di un angolo α attorno alla retta orientata r basta concatenare le seguenti trasformazioni

$$\mathrm{T}(-x_0, -y_0, -z_0)\, Rot_z(-\theta)\, Rot_y(-\phi)\, Rot_z(\alpha)\, Rot_y(\phi)\, Rot_z(\theta)\, \mathrm{T}(x_0, y_0, z_0)$$

Esempio: sia r la retta passante per i punti $P = (2, 1, 5)$ e $Q = (4, 7, 2)$, orientata secondo il vettore direzione $Q - P = (2, 6, -3)$. Il vettore $Q - P$ ha lunghezza 7 e quindi $v = \left(\frac{2}{7}, \frac{6}{7}, -\frac{3}{7}\right)$. Dunque $\cos(\phi) = -\frac{3}{7}$ e $\sin(\phi) = \frac{2\sqrt{10}}{7}$ (la colatitudine è compresa tra 0 e π, con seno positivo) e $\cos(\theta) = \frac{2}{7\sin(\phi)}$, $\sin(\theta) = \frac{6}{7\sin(\phi)}$, cioè $\cos(\theta) = \frac{1}{\sqrt{10}}$, $\sin(\theta) = \frac{3}{\sqrt{10}}$.

⊟ *matrice..*

```
> theta:=arctan(3/sqrt(10),1/sqrt(10)):phi:=arctan(2*sqrt(10)/7,-3/7):
```
la funzione **arctan** con due argomenti x,y calcola l'arcotangente di x/y e permette di controllare i segni delle funzioni seno e coseno

```
> m:=simplify(evalm(T3d(-2,-1,-5)&*Rotz(-theta)&*Roty(-phi)&*Rotz(alpha)&*Roty(phi)&*Rotz(theta)&*T3d(2,1,5)
```

```
));
>
```

Osservazione: gli angoli θ e ϕ consentono di *orientare* facilmente un oggetto tridimensionale (si veda ad esempio la finestra dei grafici 3d in Maple, dove è possibile cambiare la direzione di vista modificando θ e ϕ).

Riflessione rispetto a un piano arbitrario

Un piano nello spazio ha un'equazione cartesiana della forma $a\,x+b\,y+c\,z+d=0$, con a, b, c non tutti nulli. Il vettore $n=(a, b, c)$ è normale (cioè ortogonale) al piano. L'equazione del piano π passante per il punto $P_0=(x_0, y_0, z_0)$ e con direzione normale (a, b, c) è dunque

$$a\,(x-x_0)+b\,(y-y_0)+c\,(z-z_0)=0.$$

Per ottenere la riflessione rispetto al piano π, è sufficiente trasformare il piano in un piano coordinato, ad esempio il piano *xy*. Si può procedere come nella sezione precedente, traslando P_0 nell'origine e ruotando il vettore normale (a, b, c) con angoli θ e ϕ in modo da renderlo parallelo all'asse z. La riflessione è data quindi dalla concatenazione

$$\mathrm{T}(-x_0, -y_0, -z_0)\,Rot_z(-\theta)\,Rot_y(-\phi)\,\mathrm{S}(1, 1, -1)\,Rot_y(\phi)\,Rot_z(\theta)\,\mathrm{T}(x_0, y_0, z_0)$$

Esempio: consideriamo il piano $2\,x-y+2\,z-2=0$, passante per il punto $P_0=(1, 0, 0)$. Il vettore normale $n=(2, -1, 2)$ ha lunghezza 3 e quindi $\cos(\phi)=\frac{2}{3}$, $\sin(\phi)=\frac{\sqrt{5}}{3}$, c
os$(\theta)=\frac{2}{3\sin(\phi)}=\frac{2}{\sqrt{5}}$, $\sin(\theta)=-\frac{1}{3\sin(\phi)}=-\frac{1}{\sqrt{5}}$.

matrice..

```
> theta:=arctan(-1/sqrt(5),2/sqrt(5)):phi:=arctan(sqrt(
  5)/3,2/3):
> m:=simplify(evalm(T3d(-1,0,0)&*Rotz(-theta)&*Roty(-ph
  i)&*S3d(1,1,-1)&*Roty(phi)&*Rotz(theta)&*T3d(1,0,0)))
  ;
>
```

Esempi

```
> with(plottools):
> colore:=[red,green,blue,white]:
> tetraedro:= proc(l) local i,ll; global faccia;
  ll:=convert(l,listlist):for i from 1 to 4 do
  faccia[i]:=polygon(subsop(i=NULL,ll),color=colore[i])
  od; RETURN(convert(faccia,set)) end:
> disegno3d:=f->plots[display](f,scaling=CONSTRAINED,axes
  =NORMAL,labels=[x,y,z]):
> t0:=[[0,0,0],[1,0,0],[0,1,0],[0,0,1]]:
> disegno3d(tetraedro(t0));
```




```
> t1:=trasforma(t0,T3d(1,0,1)):traslazione
> disegno3d(tetraedro(t0) union tetraedro(t1));
```




```
> disegno3d(tetraedro(t0) union
  tetraedro(trasforma(t1,S3d(2,2,1))));cambiamento di scala
```




```
> disegno3d(tetraedro(t0) union
  tetraedro(trasforma(t0,Roty(2*Pi/3))));
```
rotazione di $\frac{2\pi}{3}$ attorno all'asse y




```
> disegno3d(tetraedro(t0) union
  tetraedro(trasforma(t0,m)));
```
riflessione rispetto al piano $2x - y + 2z - 2 = 0$



>

3.3 Rette e piani in coordinate omogenee

In coordinate cartesiane un piano ha un'equazione della forma $a\,x + b\,y + c\,z + d = 0$ (con a, b, c non tutti nulli). In coordinate omogenee, sostituendo $x = \frac{X}{W}$, $y = \frac{Y}{W}$, $z = \frac{Z}{W}$ e moltiplicando per W, si ottiene

$$a\,X + b\,Y + c\,Z + d\,W = 0.$$

Come per le rette del piano, possiamo considerare la quaterna $N = (a, b, c, d)$ come *vettore delle coordinate omogenee del piano*, e riscrivere l'equazione omogenea come prodotto scalare (in R^4)

$$P\,.\,N = 0$$

con $P = (X, Y, Z, W)$ vettore delle coordinate omogenee del punto.

Dati tre punti non allineati P_1, P_2, P_3, l'unico piano che li contiene ha coordinate omogenee $N = (a, b, c, d)$ che soddisfano le tre condizioni di ortogonalità

$$P_1\,.\,N = 0,\quad P_2\,.\,N = 0,\quad P_3\,.\,N = 0$$

Un vettore N con queste proprietà può essere ottenuto sviluppando rispetto alla prima riga il seguente determinante:

$$N = \det\left(\begin{bmatrix} e_1 & e_2 & e_3 & e_4 \\ X_1 & Y_1 & Z_1 & W_1 \\ X_2 & Y_2 & Z_2 & W_2 \\ X_3 & Y_3 & Z_3 & W_3 \end{bmatrix}\right)$$

dove $\{e_1, e_2, e_3, e_4\}$ è la base canonica di R^4: $e_1 = (1, 0, 0, 0)$, $e_2 = (0, 1, 0, 0)$, $e_3 = (0, 0, 1, 0)$, $e_4 = (0, 0, 0, 1)$. Infatti le 4 componenti a, b, c, d del vettore così ottenuto sono i complementi algebrici ottenuti dalla prima riga; sviluppando ancora rispetto alla prima riga il determinante (nullo!)

$$\det\left(\begin{bmatrix} X_1 & Y_1 & Z_1 & W_1 \\ X_1 & Y_1 & Z_1 & W_1 \\ X_2 & Y_2 & Z_2 & W_2 \\ X_3 & Y_3 & Z_3 & W_3 \end{bmatrix}\right)$$

si ha $X_1\, a + Y_2\, b + Z_1\, c + W_1\, d = 0$, cioè la prima condizione $P_1 \,.\, N = 0$. Analogamente si ottengono $P_2 \,.\, N = 0$ e $P_3 \,.\, N = 0$ prendendo matrici con prima riga P_2 o P_3. Possiamo esprimere quanto ottenuto dicendo che il piano cercato ha equazione

$$\det\left(\begin{bmatrix} X & Y & Z & W \\ X_1 & Y_1 & Z_1 & W_1 \\ X_2 & Y_2 & Z_2 & W_2 \\ X_3 & Y_3 & Z_3 & W_3 \end{bmatrix}\right) = 0$$

Esempio: il piano passante per i punti $A = (5, 4, 2)$, $B = (-1, 7, 3)$, $C = (2, -2, 9)$ ha equazione omogenea

$$\det\left(\begin{bmatrix} X & Y & Z & W \\ 5 & 4 & 2 & 1 \\ -1 & 7 & 3 & 1 \\ 2 & -2 & 9 & 1 \end{bmatrix}\right) = 0,$$

cioè $27\, X + 39\, Y + 45\, Z - 381\, W = 0$ e in coordinate cartesiane $27\, x + 39\, y + 45\, z - 381 = 0$.

In maniera analoga si possono ottenere le coordinate omogenee del punto di intersezione di 3 piani di coordinate omogenee N_1, N_2, N_3, che devono soddisfare le condizioni

$$P \,.\, N_1 = 0, \quad P \,.\, N_2 = 0, \quad P \,.\, N_3 = 0$$

Per le rette nello spazio è ancora possibile introdurre coordinate omogenee, ma in maniera più complicata e adoperando 6 coordinate omogenee. Si osservi che i punti della retta passante per i punti di coordinate omogenee $P = (X_1, Y_1, Z_1, W_1)$ e $Q = (X_2, Y_2, Z_2, W_2)$ hanno coordinate omogenee della forma

$$\alpha\, P + \beta\, Q, \quad \text{con } \alpha \text{ e } \beta \text{ parametri reali.}$$

Infatti, dalle equazioni parametriche

$$x = x_1 + t\,(x_2 - x_1),\ y = y_1 + t\,(y_2 - y_1),\ z = z_1 + t\,(z_2 - z_1),$$

sostituendo $t = \dfrac{\beta}{\alpha + \beta}$ e quindi $1 - t = \dfrac{\alpha}{\alpha + \beta}$, si ottiene

$$x = \frac{\alpha\, x_1 + \beta\, x_2}{\alpha + \beta},\ y = \frac{\alpha\, y_1 + \beta\, y_2}{\alpha + \beta},\ z = \frac{\alpha\, z_1 + \beta\, z_2}{\alpha + \beta}$$

e quindi

$$(\alpha + \beta)\,(x, y, z, 1) = \alpha\,(x_1, y_1, z_1, 1) + \beta\,(x_2, y_2, z_2, 1)$$

ed essendo le coordinate omogenee, anche $(X, Y, Z, W) = \alpha\, P + \beta\, Q$.